# $Id: PKGBUILD 183067 2013-04-17 06:14:35Z andrea $
# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=telepathy-kde-common-internals
_pkgname=ktp-common-internals
pkgver=0.6.1
pkgrel=1
pkgdesc="Common components for KDE-Telepathy"
arch=('i686' 'x86_64')
url="http://community.kde.org/Real-Time_Communication_and_Collaboration"
license=('GPL')
depends=('kdelibs' 'telepathy-logger-qt' 'telepathy-mission-control')
makedepends=('cmake' 'automoc4' 'boost' 'doxygen')
install=${pkgname}.install
source=("http://download.kde.org/stable/kde-telepathy/${pkgver}/src/${_pkgname}-${pkgver}.tar.bz2")
sha1sums=('47f99b56e2faa6ef7c760f2112a7b10445f9c709')

build() {
  mkdir build
  cd build
  cmake ../${_pkgname}-${pkgver} \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release
  make
}

package() {
  cd build
  make DESTDIR="${pkgdir}" install
}
